

CHAPTER 27

The GSLIDE Procedure

<i>Overview</i>	959
<i>About Text Slides</i>	959
<i>About Annotate Output</i>	960
<i>Procedure Syntax</i>	960
<i>PROC GSLIDE Statement</i>	961
<i>Examples</i>	963
<i>Example 1: Producing Text Slides</i>	963
<i>Example 2: Displaying Annotate Graphics</i>	965

Overview

The GSLIDE procedure produces graphics output that consists of text and straight lines that are generated by TITLE, FOOTNOTE, and NOTE statements. In addition, the procedure provides an easy way to add titles, notes, and footnotes to output that is produced entirely with an Annotate data set.


The GSLIDE procedure is useful for creating text slides for presentations. You can also overlay text slides on other graphics output with the GREPLAY procedure.

About Text Slides

Text slides contain text and graphics that are generated by SAS/GRAPH statements. To display an external text file as graphics output, use the GPRINT procedure.

Figure 27.1 on page 959 shows a slide containing text that was produced with TITLE, FOOTNOTE, and NOTE statements.

Figure 27.1 Text Slide Produced by the GSLIDE Procedure (GR27N01)




The program for this slide is in Example 1 on page 963.

About Annotate Output

Annotate output is generated by commands that are stored in an Annotate data set. Use the GSLIDE procedure to display Annotate output when you want to include TITLE and FOOTNOTE statements on the output and use certain graphics options such as BORDER. To display Annotate graphics without these, use the GANNO procedure. See Chapter 10, “The Annotate Data Set,” on page 403 for more information on creating and displaying Annotate data sets.

Figure 27.2 on page 960 shows output from an Annotate data set that is displayed with titles and footnotes that were generated by TITLE and FOOTNOTE statements.

Figure 27.2 Output from an Annotate Data Set Displayed with the GSLIDE Procedure (GR27N02)



The program for this slide is in Example 2 on page 965.

Procedure Syntax

Requirements: At least one of these is required: a TITLE, FOOTNOTE, or NOTE statement; an appearance option; the BORDER graphics option.

Global statements: FOOTNOTE, TITLE

Reminder: The procedure can include the SAS/GRAPH NOTE statement.

Supports: RUN-group processing Output Delivery System (ODS)

PROC GSLIDE <option(s)>;

PROC *GSLIDE* Statement

Creates a text slide. Optionally, it provides a border, specifies annotation, and assigns an output catalog. This is the only statement in the procedure.

Syntax

PROC *GSLIDE* *<option(s)>*;

option(s) can be one or more options from any or all of the following categories:

- appearance options:
 - ANNOTATE=*Annotate-data-set*
 - BORDER
 - CFRAME=*frame-color*
 - FRAME
 - LFRAME=*line-type*
 - WFRAME=*n*
- description options:
 - DESCRIPTION=*'entry-description'*
 - GOUT=*<libref.>output-catalog*
 - NAME=*'entry-name'*
- HTML option:
 - <IMAGEMAP=output-data-set>*

Options

You can specify as many options as you want and list them in any order.

ANNOTATE=*Annotate-data-set*

ANNO=*Annotate-data-set*

specifies a data set that includes Annotate variables that identify graphics commands and parameters.

See also: Chapter 10, “The Annotate Data Set,” on page 403

Featured in: Example 2 on page 965

BORDER

draws a border around the graphics output area, which includes the title area, the footnote area, and the procedure output area. A color specification for the border is searched for in the following order:

- 1 the CTITLE= option in a GOPTIONS statement
- 2 the CTEXT= option in a GOPTIONS statement
- 3 the default, the first color in the colors list.

See also: “Drawing Frames and Borders” on page 962

Featured in: Example 1 on page 963

CFRAME=*frame-color*

draws a frame around the procedure output area in the specified color. If you use both the CFRAME= and FRAME options, FRAME is ignored.

Note: CFRAME= does not color the background of the slide. Δ

See also: “Drawing Frames and Borders” on page 962

Featured in: Example 1 on page 963

DESCRIPTION=’*entry-description*’

DES=’*entry-description*’

specifies the description of the catalog entry for the chart. The maximum length for *entry-description* is 40 characters. The description does not appear on the chart. By default, the GSLIDE procedure assigns the description OUTPUT FROM PROC GSLIDE.

FRAME

draws a frame around the procedure output area. By default, the frame color is the first color in the colors list. If you want to specify a different color for the frame, use the CFRAME= option instead.

See also: “Drawing Frames and Borders” on page 962

GOUT=<*libref*>*output-catalog*

specifies the SAS catalog in which to save the graphics output produced by the GSLIDE procedure. If you omit the *libref*, SAS/GRAPH looks for the catalog in the temporary library called WORK and creates the catalog if it does not exist.

See also: “Storing Graphics Output in SAS Catalogs” on page 49

IMAGEMAP=*output-data-set*

creates a SAS data set that contains information about the graph and about areas in the graph. This information includes the shape and coordinates of the areas, and is used to build an HTML file that links the graph areas to other files or images. This linking provides drill-down functionality on the graph. The Imagemap data set also contains the information that is stored in the HTML variable in the Annotate data set. Therefore, in order to use IMAGEMAP= to create an HTML file, you must also use the HTML variable in the Annotate data set.

See also: “Customizing Web Pages for Drill-down Graphs” on page 100 and “HTML Variable” on page 462

LFRAME=*line-type*

specifies the line type for a frame and draws a frame around the procedure output area. Values for *line-type* are 1 through 46. Line types are shown in Figure 8.22 on page 249. By default, LFRAME=1, which produces a solid line.

NAME=’*entry-name*’

specifies the name of the catalog entry for the graph. The maximum length for *entry-name* is eight characters. The default name is GSLIDE. If the specified name duplicates the name of an existing entry, SAS/GRAPH software adds a number to the duplicate name to create a unique entry, for example, GSLIDE1.

WFRAME=*n*

specifies the width of the frame where *n* is a number. The thickness of the frame increases directly with *n*, but the thickness of the line may vary from device to device. By default, WFRAME=1, which is the thinnest line. The WFRAME= option also draws the frame.

See also: “Drawing Frames and Borders” on page 962

Featured in: Example 1 on page 963

Drawing Frames and Borders

Like the BORDER option in a GOPTIONS statement, the BORDER option in the PROC GSLIDE statement draws a box around the graphics output area. However, the

border generated by the *GSLIDE* procedure remains in effect only for the duration of the procedure.

Both *BORDER* options use the color specified by the *CTITLE=* or *CTEXT=* graphics option if either of these options is used; otherwise, the border color is the first color in the colors list.

While the *BORDER* option draws a box around the graphics output area, the *FRAME* option draws a box or frame around the procedure output area. In this case, titles and footnotes are outside of the frame. (See “Procedure Output and the Graphics Output Area” on page 29 for a description of the procedure output area.) Use *FRAME* to draw a frame in the default color, line type, and width. Otherwise, use one or more of the *CFRAME=*, *LFRAME=*, or *WFRAME=* options.

You can specify a colored frame with the *CFRAME=* option. Note that *CFRAME=* does not fill the procedure output area with color. However, you can use the *CBACK=* graphics option to provide a background color for the graphics output area. You can specify the type of line for the frame with the *LFRAME=* option and the width of the frame with the *WFRAME=* option.

Using Data-dependent Coordinates

If you use the *GSLIDE* procedure with Annotate data sets that contain data-dependent coordinates, the resulting coordinate values may exceed the range of 0 to 100 used by the graphics output area, and some of the output may not be displayed. In this case, use the *GANNO* procedure, which can scale the output to fit the available space. See also Chapter 12, “The *GANNO* Procedure,” on page 503 for details.

Using RUN Groups

Although the *GSLIDE* procedure has no action statements, it can use *RUN*-group processing to display all currently defined titles, footnotes, and notes, as well as specified annotation, each time you submit a *RUN* statement. *TITLE* and *FOOTNOTE* statements that are defined while the *GSLIDE* procedure is active remain in effect after the procedure ends. *NOTE* definitions remain in effect until the *GSLIDE* procedure ends, at which time they are canceled. To cancel *NOTE* definitions while the procedure is active, specify *RESET=NOTE* in a *GOPTIONS* statement or submit a null *NOTE* statement. See “*RUN*-Group Processing” on page 28 for details.

Examples

Example 1: Producing Text Slides

Procedure features:


PROC *GSLIDE* options:

BORDER
CFRAME=
WFRAME=

Other features:

NOTE statement

Sample library member: GR27N01



This example uses FOOTNOTE, NOTE, and TITLE statements to produce a text slide. PROC GSLIDE statement options add both a border and a frame.

Set the graphics environment.

```
goptions reset=global gunit=pct cback=blue
          colors=(white) ftext=swissb htitle=12 htext=4;
```

Define titles and footnotes.

```
title color=pink 'New Directions';
footnote1 j=1 ' ABC Engineering, Inc';
footnote2 j=1 ' January 1998'
          j=r h=3 f=swiss 'GR27N01 ';
```

Generate the slide and define additional text. BORDER draws a box around the entire graphics output area. CFRAME= draws a red box around the procedure output area. WFRAME= specifies the thickness of the frame. The first NOTE statement, which has no text, simply leaves a large blank line above the text specified by the second NOTE statement. The second JUSTIFY= causes a line break.

```
proc gslide border
          cframe=red
          wframe=4;
  note height=20;
  note height=10
        justify=center 'Goals and strategies'
        justify=center 'for the coming year';
run;
quit;
```

Example 2: Displaying Annotate Graphics

Procedure features:


PROC *GSLIDE* option:

ANNOTATE=

Other features:

Annotate data set

Sample library member: GR27N02



In this example, the *GSLIDE* procedure displays Annotate graphics along with current *TITLE* and *FOOTNOTE* definitions. See Chapter 10, "The Annotate Data Set," on page 403 for information on creating Annotate data sets.

Set the graphics environment.

```
goptions reset=global gunit=pct cback=white
        colors=(black blue green red)
        ftitle=swissb htitle=6 ftext=swiss htext=3;
```

Create the Annotate data set, ART. ART contains the commands that draw the design of triangles.

```
data art;
  length function color style $ 8;
  input function $ x y color $ style $;
  xsys='5'; ysys='5';
  datalines;
poly      30  20  blue    solid
polycont  50  20  .       .
polycont  40  50  .       .
poly      50  20  green   xl
polycont  70  50  .       .
```

```

polycont 60 50 . .
poly      40 50 red 11
polycont 60 50 . .
polycont 50 80 . .
;

```

Define title and footnotes displayed by the procedure. FOOTNOTE statements 4 and 5 have no text and are angled vertically to add space on the left and right sides between the border of the output and the frame that surrounds the procedure output area.

```

title 'Number 17';
footnote1 h=4 '"Art is anything you can get away with.'';
footnote2 j=r h=4 'D. H. Benson  ';
footnote3 j=r 'GR27N02  ';
footnote4 h=3 angle=90;
footnote5 h=3 angle=-90;

```

Display the annotate graphics on the slide with the title and footnotes. The GSLIDE procedure displays the graphics elements drawn by the commands in the Annotate data set specified by the ANNOTATE= option.

```

proc gslide annotate=art
      border
      wframe=6
      cframe=red;

run;
quit;

```


The correct bibliographic citation for this manual is as follows: SAS Institute Inc., *SAS/GRAPH® Software: Reference, Version 8*, Cary, NC: SAS Institute Inc., 1999.

SAS/GRAPH® Software: Reference, Version 8

Copyright © 1999 by SAS Institute Inc., Cary, NC, USA.

ISBN 1-58025-525-6

All rights reserved. Printed in the United States of America.

U.S. Government Restricted Rights Notice. Use, duplication, or disclosure of the software by the government is subject to restrictions as set forth in FAR 52.227-19 Commercial Computer Software-Restricted Rights (June 1987).

SAS Institute Inc., SAS Campus Drive, Cary, North Carolina 27513.

1st printing, October 1999

SAS® and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

OS/2®, OS/390®, and IBM® are registered trademarks or trademarks of International Business Machines Corporation.

Other brand and product names are registered trademarks or trademarks of their respective companies.

The Institute is a private company devoted to the support and further development of its software and related services.